

THE ATLANTA COMPUTER SOCIETY PRESENTS

COCO NUTS

Alive in '95

VOL.1 NO.11 FEB. 1995

Next Month's Meeting

March 14, 1995 Executive meeting Shoney's Jimmy Carter Blvd. Norcross, Ga.
March 21, 1995 ACS Meeting at Shoney's So. Cobb Dr. Smyrna, Ga.
March 28, 1995 ACS Meeting at Shoney's Jimmy Carter Blvd. Norcross, Ga.

----------*

PRESIDENT'S PRINTOUT
By Ken Fish

The ACS COMES ALIVE IN '95!!!!

What an attention grabbing headline. For those of you who missed the last general meeting, I'll explain why we are making such a statement. The ACS BBS has added another new area to the board. This is the Linux/ BSD areas. With files, programs, commands and informatin on these systems. Both run on a minimum 386 machine and have (to my understanding) a kernal type environment like our beloved OS9. Tom Kocourek gave a small talk on the features and system requirements for these systems at the general meeting as well as telling some of the workings of the systems themselves. He has been busy uploading lots of files to our board and says that there is quite a bit of useable stuff there for people to look at.

I the CoCofest deptartment, the attending members voted to have a 'fest this year and Newton White was asked to begin the negotiations with the Northlake Holiday Inn for the rooms again. We are looking for any and all advertising ideas to announce the 'Fest earlier this year, so send in your advertising suggestions, whether pictures or straight word copy or both for consideration. We want to increase our attendence rate over last year (even though last year was a success, it would be nice to have more attendees), and in that effort we voted to expand the 'fest. We will be inviting vendors from the following areas of MS-DOSdom : Any vendors in the CoCo emulator field, any vendors in the OS9000 field, and any from the BSD/Linux field who wish to attend and show their products.

The reason for this is the fact that ACS stands for Atlanta COMPUTER Society, not Atlanta CoCo Society. If this club is to survive we NEED to add more members and one way to accomplish this is to have more usefull things to offer the DOS people (like how to make their machines run like a CoCo! I personally can't think of anything better to do with their machines! <grin>)

Does this mean that the ACS is abandoning the CoCo? I can give you an emphatic "HELL NO!!!". As long as I am able, I will continue to advocate the CoCo. Personally, it is ALL I use. I don't have anything but my beloved CoCo and the peripherals which go to it. I don't plan on getting any clone machine for myself in the near future either. We will still have the SIGS, Echos and Files on our board for the CoCo, and will be looking for more uploads in that area. We will still have the CoCo SIG meeting for the club. But, by adding these MS-DOS areas we hope to revive our flagging membership. Who knows? Maybe some of the people who are attracted by the new areas will remember they have a CoCo in the closet and break it out!

I hope all of you the best of years in '95. I think it will be the year that the ACS Comes Alive...again. Time will tell. But you can also tell me, and are invited to do so, by writing to the newsletter. We vote on subjects like these at the general meetings, and then ask for comments and responses to be sent to the newsletter. If you don't attend meetings for one reason or another you can still give your opinion in a letter here. If you don't participate then you have no-one but yourself to blame if you don't like the direction the club is going. Even out of state members opinions are important to us, but if you don't voice them somehow we don't know them. So let us know what you think, after all it is YOUR club!

'Til Later

----------*

EDITOR'S ECHO
By Russ Keller

Last month's source code for the program was changed, thanks to Tom Kocourek, to long int from float. the %f's were changed to ld for long integers. I bought the book Teach Yourself C in 21 days and long integer was in there. The C-compiler from microware refers to C-programing guide by K&R. The book costs 35 dollars and is thin. Volume 1 is available from Microcenter. I may get it quickly so I have the reference as volume 2 is for microsoft C. I use the editor VED from Bob Van Der Poel for editing because I am used to using it. I can indent as I type in a C program without thinking. The above book seems to follow all rules thus far with Microware C.

K&R C-programing guide was obtained and all I have to do is STAY HOME and work my way through the books!

MODULAR PROGRAMMING Part 5
by Carl England

This month's module is another long one. But it differs from the other modules in that it can also be used as a stand-alone program. To use it as a stand alone program (suitable for adding to your utilities disk) add these lines:

```
00070        ORG     386
00080        JMP     DFAULT
00090        ORG     $E00
```

What does it do? It configures your system to match the needs of your printer, drives and monitor.

Lines 240 thru 440 are where you set up your system's defaults. I purposely put the defaults at the beginning of the module so that they would be easy to locate and modify if necessary.

The palette colors are only active on a CoCo3. There are two BAUD rate columns. One is for slow (normal) CPU speed and the other is for double speed. High speed operation is only available on the CoCo3 because running the CoCo1 or CoCo2 at double speed makes the display unreadable. There are four step rates available for your drives. Most drives operate best at 6ms. If yours doesn't work try the next slower rate. Even though the DISK BASIC ROM only supports 35 tracks on your drive, most drives are actually 40 (sometimes 42 but no guarantees) track drives. If you are using a 730K drive (like a 3 1/2 inch drive), you have 80 (82?) tracks available to you. Also, unless you are using a real dinosaur of a drive, you probably have a double-sided drive. If you set your system for double sided drives, then the back of drive 0 becomes drive 2, and the back of drive 1 becomes drive 3.

Lines 500 thru 620 check to see if you are using a CoCo3. If you are not, the ROMs are copied into RAM and you are switched to all RAM mode so that patches to the code can be made.

Lines 640 thru 730 set the text screen colors on the CoCo3. The 32-, 40- and 80-column screens are all set to the same colors. The border matches the background.

Lines 750 and 760 set the printer Baud rate.

The remainder fo the program makes the necessary patches to DISK BASIC. The DISK version number is checked to ensure that the correct tables are used for patching the ROM. (DISK BASIC 1.1 was a complete re-write and almost none of the routines are at the same locations as the original DISK BASIC 1.0).

Because larger disk space

requires more memory for the File Allocation Table (FAT), I chose to relocate it to high RAM. That meant that the RAM normally used for the FAT was available and so I chose to put the patches there.

The fast CPU patch works well with your printer (set the baud rate at half normal as note in module). Using cassette with this patch installed allows you to CSAVE and CLOAD programs at 3000 BAUD! It also gives reliable disk operation using all functions except the DOS command. DO NOT use high speed when running another operating system unless you know it is compatable. Using it with DISK EDTASM (or any program that loads with RUN"DOS") will probably result in unreliable operation.

The "Head-Banger Bug" which is sometimes encountered when running double-sided drives (especially when the drive masks are just poked into memory) has been eliminated.

Listing

```
00100 ************************************************
00110 *                                              *
00120 *                    MODULE: DEFAULT           *
00130 *                                              *
00140 * SETS SYSTEM DEFAULTS                         *
00150 *                                              *
00160 * INPUT CONDITIONS:  NONE                      *
00170 * OUTPUT CONDITIONS: PRINTER BAUD RATE SET     *
00180 * DRIVE TRACKS, SIDES & STEP RATE SET          *
00190 * CPU SPEED SET (COCO3 ONLY)                   *
00200 * SCREEN COLORS SET (COCO3 ONLY)               *
00210 *                                              *
00220 ************************************************
00230
00240 PALLET  FDB    $3F0C     FOREGROUND/BACKGROUND
00250 *                        COLORS RANGE: 00 THRU 3F
00260 BAUD    FCB    18        2400 BAUD
00270 *     BAUD:  300 SLOW = 180 FAST = ---
00280 *             600         87        180
00290 *            1200         41         87
00300 *            1800         25        ---
00310 *            2000         23        ---
00320 *            2400         18         41
00330 *            3600         10         25
00340 *            4800          7         18
00350 *            7200          3         10
00360 *            9600          1          7
00370 *           19200        ---         1
00380 SPEED   FCB    255       0=SLOW, 255=FAST
00390 STEP    FCB    0         6 MS
00400 *              1         12 MS
00410 *              2         20 MS
00420 *              3         30 MS
00430 TRACKS  FCB    40        40 TRACK
00440 SIDES   FCB    2         2=DOUBLE, 1=SINGLE
00450
00460
```

```
00470
00480 **** TEST FOR COCO2 OR COCO3 ****
00490 **** IF COCO2 THEN SWITCH TO RAM ****
00500 DFAULT  LDA     $E2BD
00510         CMPA    #50
00520         BEQ     DF2
00530         ORCC    #$50
00540         LDX     #$8000
00550 DF1     CLR     $FFDE
00560         LDD     ,X
00570         CLR     $FFDF
00580         STD     ,X++
00590         CMPX    #$E000
00600         BNE     DF1
00610         ANDCC   #$AF
00620         BRA     DF3
00630 **** SET UP DEFAULT SCREEN COLORS FOR COCO3 ****
00640 DF2     LDD     PALLET
00650         STD     $FFBC
00660         STA     $FFB8
00670         STB     $FFB0
00680         STB     $FF9A
00690         STB     $E035
00700         STB     $E03E
00710         STB     $E047
00720         STB     $E073
00730         STB     $E07C
00740 **** SET BAUD RATE ****
00750 DF3     LDA     BAUD
00760         STA     150
00770 **** TRANSFER PATCHES TO LOW RAM ****
00780         LDU     #DF10
00790         LDX     #$800
00800         LDB     #$2C
00810 DF4     LDA     ,U+
00820         STA     ,X+
00830         DECB
00840         BNE     DF4
00850 **** MUST USE SEPARATE TABLES FOR DISK X.0 AND X.1 ****
00860         LDX     #DF13
00870         LDA     $C155
00880         CMPA    #49
00890         BNE     DF5
00900         LDX     #DF15
00910 **** PATCH DIRECTORY TO ALLOW 158 GRANULES ****
00920 DF5     LDD     ,U++
00930         STD     [,X++]
00940         LDD     ,U++
00950         STD     [,X++]
00960         LDD     ,U++
00970         STD     [,X++]
00980 **** PATCH DIRECTORY TO ALLOW 128 FILES ****
00990         LDA     #18
```

```
01000           STA     [,X++]
01010 **** PATCH STEP RATE ****
01020           LDA     STEP
01030           STA     [,X++]
01040           ADDA    #$14
01050           STA     [,X++]
01060           ADDA    #$3C
01070           STA     [,X++]
01080 **** PATCH NUMBER OF TRACKS ****
01090           LDA     TRACKS
01100           DECA
01110           STA     [,X++]
01120           INCA
01130           STA     [,X++]
01140           STA     [,X++]
01150           STA     [,X++]
01160 **** PATCH NUMBER OF GRANULES ****
01170           DECA
01180           ASLA
01190           STA     [,X++]
01200           STA     [,X++]
01210           STA     [,X++]
01220           STA     [,X++]
01230           STA     [,X++]
01240           STA     [,X++]
01250           STA     [,X++]
01260 **** PATCH FAT LENGTH ****
01270           SUBA    #$44
01280           ASLA
01290           ADDA    #$4A
01300           STA     [,X++]
01310           CLR     $76
01320           STA     $77
01330 **** MOVE FAT TO LARGER MEMORY AREA ****
01340           LDD     #$DA00
01350           STD     [,X++]
01360           LDD     #$DA00
01370           TFR     D,U
01380           ADDD    $76
01390           ADDD    $76
01400           ADDD    $76
01410           ADDD    $76
01420           STD     $76
01430 DF6       CLR     ,U+
01440           CMPU    $76
01450           BNE     DF6
01460 **** PATCH DOUBLE SIDED DRIVES ****
01470           LDA     SIDES
01480           CMPA    #2
01490           BNE     DF7
01500           LDU     #DF11
01510           LDD     ,U++
01520           STD     [,X++]
```

```
01530           LDD     ,U++
01540           STD     [,X++]
01550           STD     [,X++]
01560           LDD     ,U++
01570           STD     [,X++]
01580           STD     [,X++]
01590           STB     [,X++]
01600           STB     [,X++]
01610 **** PATCH HIGH SPEED CPU ****
01620 DF7       LDA     SPEED
01630           BEQ     DF9
01640           LDA     $E2BD
01650           CMPA    #50
01660           BNE     DF9
01670           LDX     #DF14
01680           LDA     $C155
01690           CMPA    #49
01700           BNE     DF8
01710           LDX     #DF16
01720 DF8       LDU     #DF12
01730           LDD     ,U++
01740           STD     [,X++]
01750           LDD     ,U++
01760           STD     [,X++]
01770           LDD     ,U++
01780           STD     [,X++]
01790           LDA     ,U+
01800           STA     [,X++]
01810           LDD     ,U++
01820           STD     [,X++]
01830           LDA     ,U+
01840           STA     [,X++]
01850           CLR     $FFD9
01860 DF9       RTS
01870 **** PATCH FOR HIGH SPEED OPERATION ****
01880 DF10      PSHS    U,Y,X,B,A
01890           LDD     8,S
01900           PSHS    D
01910           LDD     #$811
01920           STD     10,S
01930           CLR     $FFD8
01940           LDA     #5
01950           RTS
01960           CLR     $FFD9
01970           RTS
01980           STA     $FF48
01990           CLR     $FFD8
02000           RTS
02010 **** 158 GRANULES ALLOWED PATCH ****
02020           PSHS    A
02030           CLRA
02040           LDA     D,X
02050           COMA
```

```
02060           PULS    A,PC
02070 **** HEAD-BANGER BUG PATCH ****
02080           LDX     #$97E
02090           LDB     $EB
02100           ANDB    #1
02110           RTS
02120 **** 158 GRANULES ALLOWED PATCH ****
02130           JSR     $81C
02140           FDB     $2730
02150           NOP
02160 **** DOUBLE SIDED DRIVES PATCH ****
02170 DF11      FDB     $4142
02180           JSR     $824
02190           NOP
02200 **** HIGH SPEED CPU PATCH ****
02210 DF12      JSR     $800
02220           NOP
02230           NOP
02240           NOP
02250           NOP
02260           JSR     $815
02270 **** PATCH TABLES FOR DISK BASIC X.0 ****
02280 DF13      FDB     $C798
02290           FDB     $C79A
02300           FDB     $C79C
02310           FDB     $C6A5
02320           FDB     $D6CD
02330           FDB     $D723
02340           FDB     $D526
02350           FDB     $D446
02360           FDB     $D572
02370           FDB     $D595
02380           FDB     $D1B0
02390           FDB     $C708
02400           FDB     $C78B
02410           FDB     $C7A0
02420           FDB     $C7BF
02430           FDB     $CC4C
02440           FDB     $CDD9
02450           FDB     $D35F
02460           FDB     $C72A
02470           FDB     $C72D
02480           FDB     $D7AC
02490           FDB     $D6C5
02500           FDB     $D70C
02510           FDB     $D6C7
02520           FDB     $D70E
02530           FDB     $D6C9
02540           FDB     $D710
02550 DF14      FDB     $D66C
02560           FDB     $D66E
02570           FDB     $D699
02580           FDB     $D69B
```

```
02590           FDB     $D50B
02600           FDB     $D50D
02610 **** PATCH TABLES FOR DISK BASIC X.1 ****
02620 DF15      FDB     $C7C8
02630           FDB     $C7CA
02640           FDB     $C7CC
02650           FDB     $C6D2
02660           FDB     $D7C0
02670           FDB     $D816
02680           FDB     $D613
02690           FDB     $D534
02700           FDB     $D65F
02710           FDB     $D682
02720           FDB     $D29D
02730           FDB     $C735
02740           FDB     $C7BB
02750           FDB     $C7D0
02760           FDB     $C7EF
02770           FDB     $CD26
02780           FDB     $CEB5
02790           FDB     $D44D
02800           FDB     $C75A
02810           FDB     $C75D
02820           FDB     $D89F
02830           FDB     $D7B8
02840           FDB     $D7FF
02850           FDB     $D7BA
02860           FDB     $D801
02870           FDB     $D7BC
02880           FDB     $D803
02890 DF16      FDB     $D75F
02900           FDB     $D761
02910           FDB     $D78C
02920           FDB     $D78E
02930           FDB     $D5FB
02940           FDB     $D5FD
02950           END     DFAULT
```

Next month I'll show you how to put this module on track 34 of your disk so that you can set your system defaults by just typing DOS.
'till later.....Carl

Atlanta Computer Society
P.O. Box 80694
Atlanta, Ga. 30366

ALLEN HUFFMAN
P. O. BOX 152442
LUFKIN TX 75915

ADDRESS CORRECTION REQUESTED

OFFICERS 1995

PRESIDENT	Ken Fish	439-5117
VICE PRESIDENT	Newton White	325-5348
TREASURER	Russ Keller	436-5094
SECRETARY	Alan Dages	469-5111
NEWSLETTER EDITOR	Russ Keller	436-5094
CLUB LIBRARIAN	Terry Dodson	463-1803
ACS BBS		636-2991